

NUMMER 2

GEDRUKTE PERIODIEK verschijnt tweemaandelijks

Verantw. Uitgever : W. HERMANS HEIDE 98 3171 WESTMEERBEEK

BLADWIJZER

3	REMARK	Redactiepraatje
4	TALK	Correspondentie +++ info
5	LOOK	Variabelen in DAI-BASIC
6	LOOK-LIST	Array variabelen +++ Partituur op printer
7	-LIST	Priemgetallen-race
8	CATALOG	Grafisch tablet (Graphic artist)
9	CATALOG	"
10	CATALOG	"
11	CATALOG	"
12	CATALOG	"
13	LIST	Real Time Clock
14	LIST	Digitaal-analoog clock H.van Cooten
15	LIST	"
16	PEEK&POKE	Zcommando +++ tip
17	LIST	Formaat-listing
18	READ	Marktoverzicht hobbycomputer-tijdschriften(1)
19	READ	"
20	READ	"
21	READ	"
22	LOOK	handboek-buggies
23	CATALOG	Programmabibliotheek *
24	CATALOG	Programmabibliotheek ** ***
25	CATALOG	Handleiding Disassembler
26	CATALOG	"
27	LOOK	Benchmarks
28	LIST	List-print J.Dessart
29	LIST	"

REMARK

DAInamic
DAI USERS CLUB

WESTMEERBEEK NOV/80

BESTE DAIInamic-LEDEN,

DE WEERSOMSTANDIGHEDEN ZIJN MET ONS: LANGE WINTERAVONDEN, HET TUINWERK IS AFGELOPEN EN HET CONTROLELAMPJE VAN DE DAIpc VERSPREIDT ZIJN WARME GLOED DOOR DE HUISKAMER... IN TEGENSTELLING MET SOMMIGE GERUCHTEN ZITTEN WIJ MET DE LIDGELDEN NOG NIET IN BRAZILIE EN GAAN WE IJVERIG DOOR MET U PROGRAMMA'S EN INFORMATIE TE BEZORGEN. ANDERMAAL ONZE DANK VOOR DE INGESTUURDE BIJDRAGEN : BRUNO VAN ROMPAEY, JEAN DESSART, JOS SCHEPENS, DONALD PHILIPPART, H. BAKKER, I. BROEKMAN, H. KOOTEN, K. ESVELD, G. CLEMENS, VAN DE SNEPSCHEUT M..... DROEGEN HUN STEENTJE BIJ.

DAInamic GAAT EEN VAARTJE, ZIJN NAAM WAARDIG : EIND OKTOBER BEHANDELDEN WIJ AL 220 AANVRAGEN VOOR LIDMAATSCHAP, UIT DE VELE REACTIES KUNNEN WIJ BESLUITEN DAT IEDEREEN ERG GELUKKIG IS MET ZIJN DAIpc. VELE LEDEN ZIJN BEZIG MET ERG INTERESSANTE PROJECTEN: MODEL TREINBESTURING, SCHAAKPROGRAMMA, GRAFIEKEN IN MACHINETAAL, VERKENNING VAN DE BASIC-ROMS, SPRAAKSYNTHESE EN MEER BOEIENDE TOEPASSINGEN. ZO ZIJN WE ER VAN OVERTUIGD DAT ER BINNEN EEN PAAR MAANDEN NOG EEN AANTAL GEWELDIGE PROGRAMMA'S UIT DE BUS EN IN DE NEWSLETTERS GAAN KOMEN.

EEN OLIESJEIK HEEFT 300 DAI'S GEKOCHT VOOR NIEUWJAARS-CADEAUTJES... OF.... HOE ZIT HET MET DE OLIEPRIJZEN???

WAAR DE TIJD HET ONS TOELIET HEBBEN WE ARTIKELS GEMAAKT MET DE SCHRIJFMACHINE, TOCH BETER LEESBAAR DAN DE MATRIX-PRINTER. OP 23 OKT JL DEDEN WIJ ONS EERSTE PUBLIEKE OPTREDEN TE TURNHOUT, ER WAS RUIME BELANGSTELLING, WE HEBBEN ER VERSCHILLENDE LEDEN KUNNEN ONTMOETEN.

ONDERTUSSEN IS OOK DAIInamic IN DUITSLAND GESTART, WIJ WENSEN FRANK EN JOACHIM EN ALLE DUITSE CLUBLEDEN VEEL SUCCES.

REST ONS NOG U HET BESTE TE WENSEN VOOR DE KOMENDE FEESTDAGEN,.....TOT IN 1981 !

GROETEN VAN JEF, JUL, FREDDY, ROGER, BRUNO, JOS, LUDO, DIRK, WILFRIED.

TALK

De heer DE DAUW A. zoekt een oplossing voor aansluiting van een CENTRONICS 779 printer via de DCE-bus. Kan iemand hiervoor de oplossing bezorgen, zodat de heer DE DAUW hardcopies kan maken ?

In laatste instantie verbetert de heer J. van Cooten zijn priemgetallen-record:

```
1 PRINT "START": FOR N = 1 TO 1000: FOR K = 2 TO N/2: M = N/K:
  IF M = 1 THEN 2: IF INT(M) = M THEN 3 : NEXT K
2 PRINT N,
3 NEXT N : PRINT "END"
```

snelheid : 491 seconden.

NORDMENDE brengt een kleuren TV op de markt met videoingang standaard en RGB moduul verkrijgbaar. Onder de naam TP 4132 biedt dit toestel een 56cm scherm, afstandsbediening, 99 kanalen. De inbouw van het RGB moduul zou 200 a 300 gulden kosten.

Een GRAFISCH programma werkt aanzienlijk vlugger met INTEGER variabelen
Een goede gewoonte : begin elk programma met:

```
MODE Ø : PRINT CHR&(12);
```

4 COLOR MODE is veel sneller dan 16 COLOR MODE.

een paar afspraken of voorstellen: Schrijfmachines beschikken niet over alle symbolen van BASIC. We stellen het volgende voor:

& staat voor STRING vb A&="NACHTRUST"

é staat voor HEX vbPOKEé75,Ø

De computer-nul kunnen we voorstellen door 0 + / : Ø

We hebben onderussen ook een verklaring (geen oplossing) voor het probleempje ivm de eerste characters in de EDIT-BUFFER in het programma FORMAAT-LISTING: In de eerste locaties van de EDIT-BUFFER komt telkens CHR&(C) terecht van lijn 6ØØ1Ø, daardoor wordt het eerste lijnnummer verminkt.

VARIABELEN IN DAI-BASIC

Voor variabelen-namen mogen in DAI-BASIC 1 tot 14 characters gebruikt worden, met de uitzonderlijke eigenschap dat ze ook door BASIC herkend worden.

Zo zijn AAAAAAAAAAAAAA en AAAAAAAAAAAAAAB twee verschillende variabelen. Het eerste character moet een letter zijn, de overigen letters of cijfers.

DAI-BASIC kent verschillende soorten variabelen:

NUMERIEKE VARIABELEN (GETALLEN)

A! = floating point 10 e-32 tot 10 e32 vb 2.98563
 B% = integer +/- 2 e32 vb 645364726

Bij power-on staat de Pc steeds in IMP FPT, in het nederlands: wordt expliciet verondersteld dat alle gebruikte numerieke variabelen floating point zijn, tenzij uitdrukkelijk het tegendeel aangeduid wordt.

We geven te kennen dat we integer-variabele willen door achter de naam % te plaatsen. Dit stellen we vast als we na programma-invoer IMP INT intypen. Bij de listing verschijnen dan alle numerieke variabelen met !, het kenmerk voor floating point. Bij de integer variabelen is het % teken verdwenen.

Floating point en integer variabelen worden onder een ganz verschillend formaat in het geheugen opgeslagen, wie machinetaal met BASIC wil laten samenwerken moet daar zeker rekening mee houden!

***** Het is mogelijk om een programma dat in FPT is gemaakt integraal om te zetten naar INT !!!!(en omgekeerd)

Dit is het menu:

```
*LOAD programma
*IMP FPT
*CLEAR xxxx (groter dan het programma)
*EDIT
*BREAK BREAK
*IMP INT
*POKE #135,2
*LIST
```

ALFANUMERIEKE VARIABELEN

STRINGS mogen van 0 tot 255 characters lang zijn en draagen het kenmerk \$. vb A\$

LOOK

ARRAY VARIABELEN


Deze kunnen de verschillende kenmerken dragen:
A!(X,Y,...) floating point array bij IMP FTP A(X,Y,...)
A%(X,Y,...) integer array
A\$(X,Y,...) string array

De subscripts kunnen maximaal 255 zijn.
Samenvattend kan A dus zowat alles herbergen:
A A! A% A\$ A(9,9) A!(9,9) A%(9,9) A\$(9,9).

Bij IMP FTP is A=A! en A(9,9)=A!(9,9).

LIST

```
5        CLEAR 5000 ##### MUSIC #####
10        DIM N(75.0,13.0)
20        FOR X=3.0 TO 11.0 STEP 2.0
30        FOR Y=1.0 TO 75.0
40        N(Y,X)=#C0
45        N(Y,X+1.0)=32.0
46        NEXT: NEXT
49        FOR X=1.0 TO 75.0
50        N(X,1.0)=32.0:N(X,2.0)=32.0:N(X,13.0)=32.0:NEXT
50        P=1.0
100       FOR X=13.0 TO 1.0 STEP -1.0
110        READ N#
120        N(X*4.0,P)=#D1
125        N(X*4.0+1.0,P)=32.0
130        N(X*4.0+2.0,P)=ASC(LEFT$(N#,1))
140        N(X*4.0+3.0,P)=ASC(RIGHT$(N#,1))
150        P=P+1.0
160        NEXT
200        FOR X=0.0 TO 13.0
210        FOR Y=1.0 TO 75.0
220        PRINT CHR$(N(Y,X));:NEXT:PRINT :NEXT
1000       DATA la,so,fa,mi,re,do,si,la,so,fa,mi,re,do
```



PRIME NUMBERS H. BAKKER 294sec.

```

1 PRINT CHR$(12):PRINT :PRINT " ***** PRIME NUMBERS FROM 1 TILL
1001 *****"
20 PRINT :PRINT " (ref. DAInamic newsletter 1 / INTERFACE AGE)":PR
INT
25 PRINT " EXECUTION TIME OF THIS PROGRAM = 294 SEC."
30 PRINT :PRINT " (without MATH-chip)"
35 REM *****
40 REM ALLEEN ONEVEN GETALLEN TUSSEN 1 EN 1000 WORDEN
47 REM ONDERZOCHT, OMDAT EVEN PRIEMGETALLEN NIET BESTAAN
48 REM
49 REM NA GETAL 3 WORDT ALLEEN NOG GEDEELD DOOR ONEVEN
50 REM GETALLEN GROTER DAN 2 EN KLEINER DAN DE HELFT
53 REM VAN HET GETAL WAT ONDERZOCHT WORDT.
55 REM *****
60 CURSOR 20,5:INPUT "(return)";D$
70 PRINT CHR$(12):PRINT " STARTING:"
150 FOR N=3.0 TO 1001.0 STEP 2.0:FOR K=2.0+A TO N/2.0 STEP 2.0:M=N/K:L=
INT(M):IF M<>L THEN NEXT K:PRINT N:A=1.0:NEXT N:GOTO 999
160 NEXT N
999 PRINT " **** FINISHED ****":PRINT
1000 INPUT " (return)";D$:LIST

```

Een PRIEMGETALLENOPLOSSING van de heer H.van Cooten:

```

1 PRINT CHR$(12) : PRINT "START"
2 FOR N = 1 TO 1000 : FOR K = 2 TO N/2 : M = N/K
3 IF M = 1 THEN 7
5 IF FRAC(M) = 0 THEN 8
6 NEXT K
7 PRINT N
8 NEXT N : PRINT "FINISHED" : END

```

Tijd : 568 sec. zonder mathchip!

Het programma kan sneller lopen als na de THEN geen regelnummers worden gegeven maar de instructie van de regel direct wordt uitgevoerd vb: 3 IF M = 1 THEN PRINT N : NEXT N of

5 IF FRAC(M) = 0 THEN NEXT N. Hiermee kunnen enige seconden gewonnen worden, echter zal na 997 (laatste priemgetal) een foutbericht gegeven worden : NEXT WITHOUT FOR IN LINE 6.

Een extra 10 seconden winst kan worden behaald door variabele N te doorlopen met STEP 2 : 2 FOR N = 1 TO 1000 STEP 2, echter het enige even priemgetal "2" wordt dan niet gegenereerd, ook het algoritme wordt op deze wijze aangetast, hetgeen wellicht "verboden" is.

G R A F I S C H T A B L E T

T	A	I	E	C	S	W	O	?	D:	L	D)	H	N	T	XXXX
<p>TEXT</p> <p>FRAME : 4 sides of rectangle</p> <p>FILL : fill rectangle</p> <p>DRAW : line from point to point</p> <p>TRACE : drawing continuous lines (event = on/off switch)</p> <p>ELLIPS : left/upper coordinates</p> <p>CIRCLE : mode depends on COMPUTEFLAG</p> <p>WIPE : clears a part of the picture</p> <p>DOT : dots with event</p> <p>COMPUTEFLAG : display results Y/N</p> <p>DISTANCE between 2 coordinates</p> <p>CLEAR : clears picture</p> <p>LENGTH of cont.lines</p> <p>HELP : info page</p> <p>SCANNING</p> <p>TEXT</p>															

ENTER HERE
FOR COLOR CHOSE

A

register 1 4

register 2 3

register 3 2

L D F ?

CATALOG

CATALOG

G R A F I S C H T A B L E T (Graphic Artist)

Dit programma maakt het mogelijk tekeningen, grafieken en teksten te creëren. Invoer gebeurt met de paddels (2 pot's), eventknop en het toetsenbord voor de teksten.

Het programma biedt zich aan als een tablet met bovenaan het menu. Op het tablet bevindt zich een cursor (non-destructief) die met de stand van de paddel kan gestuurd worden.

Door de cursor even in een hokje te sturen wordt die bepaalde functie gekozen.

Het programma loopt in MODE 5 of MODE 6 volgens keuze.

EVENT is de algemene ON/OFF switch voor de meeste opdrachten.

Het programma is modulair opgebouwd zodat opdrachten naar wens kunnen gewijzigd of uitgebreid worden.

De standaard opdrachten:

- T TRACE : De cursor volgt de stand van de paddel en laat een spoor na in de gekozen kleur. EVENT is ON/OFF switch.
- A DRAW : Kies eerste punt met EVENT, dan tweede punt met EVENT, de beide punten worden verbonden met een rechte lijn.
- F FILL : De standaard FILL-opdracht van DAI-BASIC. (overstaande hoekpunten bepalen met EVENT)
- E FRAME : Tekent de 4 zijden van de rechthoek (zie FILL).
Het programma begint in deze MODE.
- C CIRCLE: Kies eerst het middelpunt, daarna een willekeurig punt op de omtrek.
Er zijn verschillende mogelijkheden voor de cirkel:
in MODE 5, COMPUTEFLAG ON

Eerst wordt de omtrek getekend, daarna wordt de cirkel opgevuld met stralen. Het tekenen kan op alle momenten onderbroken worden door EVENT.

vervolg CIRCEL in MODE 5

Wordt het tekenen onderbroken tijdens het tekenen van de omtrek dan verschijnt achter D de lengte van dit deel van de omtrek. Wordt het tekenen onderbroken bij het stralentekenen, dan komt achter A de oppervlakte van dit deel van de circel.

noot: alle verkregen cijfergegevens duiden op scherpunten (250X250)
in MODE 5, COMPUTEFLAG OFF

Er wordt een volle circel getekend met horizontale lijnen zo vermijden we kleurenconflicten in deze 16kleurenMODE.
in MODE 6, COMPUTEFLAG ON

Eerst wordt de omtrek getekend, daarna de stralen.

°Tijdens het tekenen van de stralen kan met de horizontale pot COLORG beïnvloed worden, dit heeft effect op de bestaande tekening en houdt verrassende mogelijkheden in.

Voor cijferresultaten : zie MODE 5 COMPUTEFLAG ON
in MODE 6, COMPUTEFLAG OFF

Dezelfde volle circels als in MODE 5.

noot : Circels en ellipsen die buiten het tablet zouden lopen worden niet getekend, U hoort een akoestisch signaal ter verwittiging.

S ELLIPS : U bepaalt een punt LINKS op de omtrek, daarna het bovenste punt van de ELLIPS. De volgorde is belangrijk.

W WIPE : Wist een gedeelte van de bestaande tekening. Bepaal met EVENT de overstaande hoekpunten van een tekening.

O DOT : Plaatst een dot bij elke druk op EVENT.

? COMPUTE : ON/OFF switch voor berekeningen.

Onderaan rechts verschijnt een VLAG ter herinnering.

D: DISTANCE: Met COMPUTEFLAG ON verschijnt achter D de afstand tussen twee met EVENT gekozen punten. Met COMPUTEFLAG OFF zijn we in DOT MODE.

L CLEAR : Maakt de tablet schoon, we blijven in de vorige MODE, ook de PENCOLOR blijft bewaard.

D) Met COMPUTEFLAG ON krijgen we benaderende waarden van oppervlakte en omtrek van een willekeurige figuur.

De nauwkeurigheid is mede beperkt door de array-dimensie (255)

- D) Met COMPUTEFLAG OFF krijgen we de lengte van een willekeurige lijn.
- H HELP : Infopagina met beknopte informatie over de diverse opdrachten. Deze opdracht kan eventueel gewijzigd worden in een andere grafische functie. (vb: het tablet horizontaal en vertikaal vullen met referentielijnen)
- N SCANNING: Met COMPUTEFLAG OFF: verticale lijnen
Met COMPUTEFLAG ON: horizontale lijnen.
Een gekozen rechthoek (zie FILL of FRAME) wordt gescand met vert. of hor. lijnen.
De afstand tussen de lijnen wordt bepaald door de vorige opdracht.
vb.: eerst A(DRAW) dan N geeft lijnen step 2
eerst I(FILL) dan N geeft lijnen step 3
- noot : eerst T(TRACE) dan geeft ver.+hor.lijnen met vaste afstand.
- T TEXT Tekent op de plaats van de CURSOR een letter, gekozen met het toetsenbord.
- COLOR : Het laatste open hokje dient voor PENCOLORselectie. Boven dit hokje verschijnt een wit vlakje (mode5), bij verdraaiing van de hor.pot verschijnen achtereenvolgens de 15 beschikbare kleuren (achtergrond niet, want dan zitten we opgeschept met een onzichtbare CURSOR !)
Bij druk op EVENT wordt de getoonde kleur gekozen voor PENCOLOR. Een sireentje verwittigt dat we het menu moeten verlaten. De gekozen kleur wordt gedisplayed naast de tablet.

OPMERKINGEN VOOR MODE 6(4 kleuren):

De verschillende kleuren worden niet getoond maar wel de hexadecimale voorstelling:

F=15=wit E=14=geel D=13=lichtgroen
C=12=1.blauw B=11=rose A=10=oranje
enz.

KLEURENKEUZE IN MODE 6(4kleuren mode)

Indien we in MODE 6 een nieuwe kleur kiezen moet COLORG aangepast worden. Dit gebeurt bij iedere keuze volgens volgende afspraken:

COLORG 8 X X X : de eerste kleurregister (achtergrond) blijft ongewijzigd.

COLORG X K X X : we kunnen de kleur in het tweede register wijzigen indien we voordien werkten in TRACE (eerste hokje) of DRAW (tweede hokje)

COLORG X X K X : derde register wijzigen indien in FILL MODE.

COLORG X X X K : vierde register wijzigen indien in FRAME MODE.

De overige MODES wijzigen ook register 4.

Als we in MODE 6 terugkeren van het menu zijn we in FRAME MODE, behalve voor T A en I.

Daar de kleurenkeuze COLORG beïnvloedt, wordt de bestaande tekening onmiddellijk aangepast tzt: alles wat getekend is in het bewuste register neemt de nieuwe kleur aan.

OPMERKINGEN: Het ligt voor de hand dat we de geconstrueerde tekening willen bewaren. Oorspronkelijk hadden we ons voorgesteld alle tekenopdrachten volgens codes in arrays op te slaan en deze op tape te bewaren. Door geheugengebrek (48K !) moest deze mogelijkheid vervallen.

De tekening kan wel bewaard worden door een WRITE in UTILITY: W63B8 BFFF. Hiervoor moeten we echter naar MODE5A of 6A, zodat de tekening bij het laden tevoorschijn komt in de A-mode.

De hele tekening kan echter wel te voorschijn komen door in COMMAND MODE: MODE5 of MODE6: WAIT TIME 65000., indien gewenst kan de WAIT TIME in een for-next lus geplaatst worden, zodat de tekening heel lang zichtbaar blijft.

Het saven van deze ±22K duurt ongeveer 5 minuten !!

$22000 \times 8 / 600 = 293 \text{ sec} = 4.88 \text{ minuten.}$

Op tape vindt U een tekening die op deze wijze bewaard is: U kan deze tekening laden door: MODE 6/UT/R (het programma moet niet HET PROGRAMMA VEREIST 48K. vooraf geladen worden!)

LIST

```

05      CLEAR 300
60006   POKE #29C,3:POKE #29E,0:POKE #3EC,#80:POKE #3ED,#28
60010   FOR T1%=0.0 TO 10.0:READ D%
60020   FOR T1%=0.0 TO 15.0:READ D1%
60026   POKE D%,D1%:D%=D%+1.0:NEXT:NEXT
60030   POKE #71,3:POKE #70,0:REM interrupt 7 aanpassen
60040   REM eenheden seconden of #3BA
60041   REM tientallen seconden of #3BB
60042   REM eenheden minuten of #3BC
60043   REM tientallen minuten of #3BD
60044   REM eenheden uren of #3BE
60045   REM tientallen uren of #3BF
60046   REM NA "RUN" KAN U ZONDER MEER
60047   REM EEN NIEUW PROGRAMMA LADEN OF INVOEREN
60048   REM DE KLOK LOOPT NIET VERDER BIJ
60049   REM LOAD,SAVE EN CHECK.
60050   REM NA EEN RESET KAN U DE KLOK OPNIEUW STARTEN
60051   REM DOOR : POKE#71,3:POKE#70,0
60052   REM U KAN HET DISPLAYEN STOPPEN DOOR :
60053   REM POKE #363,#03 (vb INDIEN U NAAR EDIT-MODE GAAT)
60054   REM DISPLAY HERVATTEN DOOR :
60055   REM POKE #363,#02
60056   REM VOOR 32K TOESTELLEN :
60057   REM IEDER #BF.. VERVANGEN DOOR #7F..
60058   REM VOOR 12K TOESTELLEN :
60059   REM IEDER #BF.. VERVANGEN DOOR #2F..
60060   REM VOOR 8K TOESTELLEN :
60061   REM IEDER #BF.. VERVANGEN DOOR #1F..
60100   DATA #300,#05,#05,#E5,#F5,#21,#B9,#03,#06,#0A,#0E,#06,#16,#00,#1E
, #32,#34
60110   DATA #310,#7B,#BE,#C2,#57,#03,#72,#23,#34,#78,#BE,#C2,#5E,#03,#72
, #23,#34
60120   DATA #320,#79,#BE,#C2,#5E,#03,#72,#23,#34,#78,#BE,#C2,#5E,#03,#72
, #23,#34
60130   DATA #330,#79,#BE,#C2,#5E,#03,#72,#23,#34,#78,#BE,#C2,#41,#03,#72
, #23,#34
60140   DATA #340,#2B,#23,#3E,#02,#BE,#C2,#5E,#03,#2B,#3E,#04,#BE,#C2,#5E
, #03,#36
60150   DATA #350,#00,#23,#36,#00,#C3,#5E,#03,#F1,#E1,#01,#C1,#C3,#A9,#D9
, #3A,#EF
60160   DATA #360,#BF,#FE,#7A,#C2,#57,#03,#21,#BA,#03,#7E,#C6,#30,#32,#77
, #BF,#23
60170   DATA #370,#7E,#C6,#30,#32,#79,#BF,#23,#7E,#C6,#30,#32,#7D,#BF,#23
, #7E,#C6
60180   DATA #380,#30,#32,#7F,#BF,#23,#7E,#C6,#30,#32,#83,#BF,#23,#7E,#C6
, #30,#32
60190   DATA #390,#85,#BF,#3E,#20,#32,#75,#BF,#32,#7B,#BF,#32,#81,#BF,#32
, #87,#BF
60200   DATA #3A0,#C3,#57,#03,#00,#00,#00,#00,#00,#00,#00,#00,#00,#00,#00
, #00,#00
60500   INPUT "INPUT THE TIME < HHMMSS >";T$:PRINT :A%=#3BF
60510   FOR D=0.0 TO LEN(T$)-1.0:T1%=MID$(T$,D,1)
60520   IF ASC(T1%)>47 AND ASC(T1%)<58 THEN POKE A%,VAL(T1%):A%=A%-1.0:IF
A%=#3B9 THEN PRINT CHR$(12):LIST 60040-60070
60530   NEXT
60550   END

```

LIST


DIGITAAL/ANALOOG CLOCK

H. VAN COOTEN 8K

```

/      REM CLOCK PROGRAM
15     REM 1 MINUUT DIGITAAL
20     REM 1 MINUUT ANALOOG
25     REM AFWISSELEND !!
30     REM MET SECONDENWIJZER BIJ ANALOGE AANWIJZING.
35     REM
37     REM HARRY VAN COOTEN/JUNI 80/8K DAI
40     PRINT CHR$(12);
50     COLORT 12 4 12 12
60     PRINT "CLOCK PROGRAM. "
103    POKE #BFEF,#5C:REM OF #1FEF,#2FEF,#7FEF
104    PRINT
105    PRINT          (8K) (12K) (32K)
106    PRINT
107    PRINT
110    PRINT          CLOCK PROGRAM. "
111    REM
120    PRINT          HARRY VAN COOTEN. "
130    PRINT          ":D$
140    PRINT
150    PRINT
160    PRINT
170    INPUT "INVOER S. U. P.  UREN TIENTALLEN":A%
180    PRINT
185    IF A%>2.0 THEN 170
190    INPUT "INVOER S. U. P.  UREN EENHEDEN  ":U%
195    IF U%>9.0 THEN 190
200    PRINT
210    INPUT "INVOER S. U. P.  MINUTEN TIENTALLEN":B%
211    PRINT
212    INPUT "INVOER S. U. P.  MINUTEN EENHEDEN  ":M%
213    PRINT
214    INPUT "INVOER S. U. P.  SECONDEN TIENTALLEN  ":C%
215    PRINT
216    INPUT "INVOER S. U. P.  SECONDEN EENHEDEN  ":S%
217    PRINT
218    IF B%>5.0 THEN 210
219    IF M%>9.0 THEN 212
220    IF C%>5.0 THEN 214
221    IF S%>9.0 THEN 216
225    REM
260    PRINT CHR$(12);
261    PRINT "  DIGITAL CLOCK. "
262    POKE #BFEF,#5C:REM OF #1FEF,#2FEF,#7FEF
280    CURSOR 10,17
289    PRINT A%;U%; ". ";B%;M%; "    ":C%;S%
290    A=6.0
291    AA=#6B
292    POKE #BFEF-(A*#86),AA
295    S%=S%+1

```



```

296 IF S%>9 THEN 400
300 IF S%>6.0 AND C%>4.0 THEN 2040
320 GOTO 1000
400 S%=0
410 C%=C%+1
420 IF C%>5 THEN 450
430 GOTO 1000
450 C%=0
460 M%=M%+1
470 IF M%>9 THEN 500
480 GOTO 1000
500 REM
502 M%=0
510 B%=B%+1
520 IF B%>5 THEN 550
530 GOTO 1000
550 B%=0
600 U%=U%+1
610 IF U%>3 THEN 700
620 GOTO 1000
700 U%=0
710 A%=A%+1
720 IF A%>2 THEN 800
730 GOTO 1000
800 A%=0
1000 FOR U=1.0 TO 1325.0
1010 NEXT U
1015 GOTO 280
2040 REM
2050 S9=PI/30.0
2060 M9=PI/1800.0
2070 U9=PI/21600.0
2071 M%=M%+1
2072 IF M%>9 THEN 2074
2073 GOTO 2087
2074 M%=0
2076 B%=B%+1
2077 IF B%<6 THEN 2087
2078 B%=0
2079 U%=U%+1
2080 IF U%>3.0 AND A%=2.0 THEN 2084
2081 IF U%>9.0 THEN 2083
2082 GOTO 2087
2083 U%=0:A%=A%+1:IF A%<3 THEN 2087
2084 A%=0:GOTO 2087
2087 M=(B%*10.0)+M%
2088 U=(A%*10.0)+U%
2090 MODE 2:REM OF MODE 4 OF MODE 6
2110 FOR X=0.0 TO 2.0*PI STEP PI/6.0
2120 DOT 30+30*SIN(X),30+30*COS(X) 15
2130 NEXT X
2150 M3=M*PI/30.0
2160 U3=U*PI/6.0+M3/12.0
2170 S1=30.0
2171 S2=30.0
2172 M1=30.0
2173 M2=30.0
2174 U1=30.0
2175 U2=30.0
2190 FOR S3=0.0 TO 2.0*PI STEP S9
2200 DRAW 30,30 S5,S6 0
2210 DRAW 30,30 S1,S2 10
2220 DRAW 30,30 M5,M6 0
2230 DRAW 30,30 M1,M2 15
2240 DRAW 30,30 U5,U6 0
2250 DRAW 30,30 U1,U2 15
2270 WAIT TIME 27
2290 S5=S1
2292 S6=S2
2300 M5=M1
2302 M6=M2
2310 U5=U1
2312 U6=U2
2330 S1=30.0+28.0*SIN(S3)
2340 S2=30.0+28.0*COS(S3)
2350 M3=M3+M9
2360 M1=30.0+28.0*SIN(M3)
2370 M2=30.0+28.0*COS(M3)
2380 U3=U3+U9
2390 U1=30.0+20.0*SIN(U3)
2400 U2=30.0+20.0*COS(U3)
2403 NEXT S3
2404 MODE 0
2405 C%=0
2407 S%=0
2410 M%=M%+1
2420 IF M%<10 THEN 260
2430 M%=0
2440 B%=B%+1
2450 IF B%<6 THEN 260
2460 B%=0
2470 U%=U%+1
2480 IF U%<4 THEN 260
2490 IF U%<10 THEN 260
2500 U%=0
2510 A%=A%+1
2520 IF A%<3 THEN 260
2530 A%=0
2540 GOTO 260
2971 M%=M%+1

```

PEEK & POKE

MACHINE LANGUAGE UTILITY : HET Z COMMANDO

IN DE PERSONAL COMPUTER MANUEL HANDELT EEN HOOFDSTUK OVER MOGELIJKHEDEN VAN DE UTILITY. WIE ECHTER PROBEERT EEN MACHINETAALPROGRAMMA TE STARTEN MET HET G OF L COMMANDO, ZAL VOLGENS ONZE ERVARING ALS ENIG RESULTAAT EEN BONTE VERZAMELING KLEUREN OP HET SCHERM BEKOMEN. OM DEZE CRASH VAN DE MACHINE TE VERMIJDEN DIENEN WE VOORAF EEN INITIALISERING TE DOEN. DIT KAN EENVOUDIG DOOR GEBRUIK VAN HET Z COMMANDO. SPIJTIG GENOEG VERMELDT DE MANUEL OVER DIT COMMANDO EN HET GEBRUIK ERVAN VOOR G OF L COMMANDO'S NIETS. HET Z COMMANDO BESTAAT UIT DE LETTER Z ONMIDDELIJK GEVOLGD DOOR EEN CIJFER VAN 1 TOT 7. HIERONDER GEVEN WE EEN (ONVOLLEDIG) OVERZICHT :

COMMANDO	GEVOLG
21	RESET ALLE CPU REGISTERS STACKPOINTER OP F900 GEZET
22	HERSTELLEN VAN DE INTERRUPTVECTOREN (0 TOT 7) MASK, TICC EN GIC BYTES WORDEN OP BEPAALDE STAND GEZET (TE BEKIJKEN MET U COMMANDO)
23	Z1 + Z2

VOOR GEBRUIK VAN EEN G OF L COMMANDO VOEREN WE EERST EEN Z3 UIT. HIERDOOR WORDT INTERRUPTVECTOR 0 KLAAR GEZET VOOR GEBRUIK DOOR HET L COMMANDO. DEZE Z3 MOET SLECHTS 1 MAAL GEBEUREN NADAT DE MACHINE OPGEZET IS OF NA EEN RESET.

TIP ... TIP ... TIP ... TIP ... TIP ... TIP ... TIP ... TIP ...

PROGRAMMA-CHAINING

INDIEN UW PROGRAMMA TE GROOT IS VOOR HET BESCHIKBAAR GEHEUGEN EN HET PROGRAMMA KAN GESPLITST WORDEN IN DEELPROGRAMMA'S DAN KAN HET LADEN EN STARTEN VAN DE VERSCHILLENDE DELEN BIJ DAIPc AUTOMATISCH GEBEUREN. U ZORGT ERVOOR DAT DE LAATSTE LIJN VAN IEDER DEEL "LOAD" BEVAT. DEZE "LOAD" ZAL:

- 1) HET OUDE DEEL WISSEN
- 2) DE CASSETTE STARTEN
- 3) HET NIEUWE DEEL LADEN
- 4) EN ... HET NIEUWE DEEL STARTEN !!!

DEZE MOGELIJKHEID OPENT INTERESSANTE PERSPECTIEVEN, VOORAL VOOR 8K EN 12K TOESTELLEN.


```

REM *****FORMAAT-LISTING*****
  8  REM
 10  REM NEEM ZELF HET FORMAAT VAN DE LISTING IN DE HAND
120  REM DEZE PROCEDURE BESTAAT ERIN DAT WE EERST HET PROGRAMMA
130  REM NAAR DE EDIT-BUFFER STUREN,
140  REM DOOR TWEEMAAL "BREAK" HET PROGRAMMA DAAR LATEN ZITTEN,
150  REM EN DAN DE LIJNNUMMERS 60000-60050
160  REM DE EDIT-BUFFER LATEN UITPRINTEN.
170  REM IN DE EDIT-BUFFER STAAT NL NIET DE RUNTIME-CODE,
180  REM MAAR HET GEDECODEERD FORMAAT.
190  REM VOORAF MOET EEN CLEAR GEMAAKT WORDEN,
1100  REM AFHANKELIJK VAN DE LENGTE VAN HET PROGRAMMA.
1110  REM DE EDIT-BUFFER MOET NL HELEMAAL IN DE HEAP-RUIMTE KUNNEN,
1120  REM ANDERS GAAN WE DE RUNTIME-CODE OOK UITPRINTEN.
1130  REM DIT KAN GETEST WORDEN DOOR EERST DE LISTING TE LATEN
1140  REM GEBEUREN MET DE PRINTER UIT.
1150  REM ER IS GEEN BEPERKING VOOR DE LENGTE VAN DE LIJNEN.
1160  REM DE SUBROUTINE OP 60030 ZORGT ERVOOR DAT DE LIJNNUMMERS
1170  REM LOGISCH WORDEN GERANGSCHIKT.
1180  REM ALLE WIJZIGINGEN IN DE EDIT-BUFFER BLIJVEN INTACT
1190  REM ZODAT U DE SCHIKKING NAAR SMAAK KAN UITVOEREN.
10030  REM DE PROCEDURE VERLOOPT ALS VOLGT:
10035  REM CLEAR xxxxxx
10040  REM EDIT n m / (gewenst deel van het programma)
10050  REM eventuele wijzigingen in edit-buffer(schikking)
10060  REM BREAK BREAK
. 1070  REM LOAD "programma met lijnummers 60000-60050"
10080  REM U kan ook het programma intikken.
10090  REM RUN 60000
11000  REM HET EERSTE LIJNNUMMER WORDT VERMINKT,
11010  REM DOORDAT ER WAARSCHIJNLIJK EEN POINTER OPGETELD
11020  REM WORDT,MISSCHIEF VINDT IEMAND HIERVOOR DE OPLOSSING?

```

```

60000 FOR X=PEEK(#A2)+PEEK(#A3)*256 TO PEEK(#A4)+PEEK(#A5)*256-2
60002 IF FLAG=0.0 THEN FLAG=1.0:GOSUB 60030
60005 C=PEEK(X):IF CNT>75 THEN CNT=0.0:PAGE=PAGE+1.0:PRINT :PRINT "
60007 IF PAGE>=50.0 THEN PRINT :PRINT :PRINT :PAGE=0.0:GOSUB 60030
60010 PRINT CHR$(C):CNT=CNT+1.0
60015 IF C=13.0 THEN PAGE=PAGE+1.0:GOSUB 60030:CNT=0.0
60020 NEXT
60025 GOTO 60025
60030 FOR Y=1.0 TO 6.0
60035 IF PEEK(X+Y)=32 GOTO 60040
60037 NEXT
60040 FOR M=1.0 TO 7.0-Y:PRINT " ":NEXT
60050 RETURN

```

READ

Ditmaal zullen we proberen een klein overzichtje te geven van de op de markt bestaande microcomputertijdschriften. Het aanbod is echter zo groot dat het onmogelijk is een volledig overzicht te geven van wat tegenwoordig op de markt beschikbaar is.

Daarom ook hebben we ons enigszins moeten beperken in onze keuze. Wat u niet zal vinden in dit overzicht zijn:

- Vaktijdschriften
 - 6502, 6800, enz. gespecialiseerde tijdschriften
 - Algemene-electronica-tijdschriften
(zie hieronder voor een kort lijstje van die tijdschriften die een computer-supplement o.i.d. hebben)
 - Commerciële newsletters
 - Newsletters voor specifieke microprocessors of -computers
- kortom al die tijdschriften die niet onmiddellijk belangrijk zijn voor DAI-PC gebruikers. Het is duidelijk dat het niet opnemen van bepaalde tijdschriften in dit overzicht niet wil zeggen dat ze niet interessant zijn!

Er werd betracht zoveel mogelijk objectieve informatie te verzamelen. Deze is echter maar ter indicatie opgenomen. We zijn nl. niet altijd op de hoogte van de laatste adreswijzigingen of veranderingen (lees omslag) qua abonnementsprijzen. Het overzicht dat nu volgt is verre van volledig, daar zijn we ons ten volle van bewust. Daarom, indien u nog andere tijdschriften kent laat het dan A.U.B. weten zodat de andere leden van DAINAMIC er eveneens kunnen van genieten.

Voor reacties, aanvullingen, tips, e.d. kan u de volgende adres terecht: Jos Schepens
St.-Jorisvilde 53
9330 DENDERMONDE

Nog enkele algemene opmerkingen:

- De lijst is alfabetisch gerangschikt
- Hardware en software werden resp. tot hard en soft afgekort
- Wat de abonnementen betreft moet men soms opletten:
bepaalde uitgeverijen noteren u als abonnee tot wederopzegging
(een nogal vervelende politiek volgens mij).
- Sommige buitenlandse uitgeverijen quoteren geen abonnementsprijzen voor het buitenland. Het beste vraagt u dan vooraf de juiste prijs.
- Meestal wordt gevraagd te betalen in de buitenlandse munt.
- Enkele tijdschriften die regelmatig een supplement "micro" hebben
 - * Elektuur (NL.)
 - * Popular Electronics (USA)
 - * Practical Electronics (UK)
 - * Radio Bulletin (NL.)
 - * Radio Electronics (USA)
 - * Enz., Enz.

NAAM: BYTE
LAND / TAAL: USA / ENGELS
PERIODICITEIT: MAANDELIJKS
ADDR. UITGEVER: BYTE Publications Inc, 70 Main St, Peterborough NH 03458
ABONNEMENTEN: BYTE Subscriptions, PO Box 590, Martinsville NJ 08836
ABONNEMENTSPRIJZEN: \$32 (1 year airmail EUROPE)
PRIJS PER NUMMER: \$4 (EUROPE), \$2.50 (USA)
INHOUD: Hard + Soft + general features + background info
GEMIDDELDE OMVANG: 300-400 p. (laatste nr.: 400 p. !)
BEOORDELING: Reeds in zijn vijfde Jaar en "still one of the best"
zeer deselijk qua inhoud, echter weinig kant en klare BASIC-programma's

NAAM: CHIP
LAND / TAAL: DUITSLAND / DUIJS
PERIODICITEIT: MAANDELIJKS
ADDR. UITGEVER: Vogel-Verlag KG, Postfach 67 40, D-8700 Wurzburg 1
ABONNEMENTEN: CHIP Leser-Service, Vogel-Verlag, Postfach 67 40
D-8700 Wurzburg 1
ABONNEMENTSPRIJZEN: 64,-DM (Ausland)
PRIJS PER NUMMER: 4,80 DM (Ausland), 4,50 DM (Inland)
INHOUD: Soft + Tests + Background info + Tutorials
GEMIDDELDE OMVANG: 100-150 p.
BEOORDELING: Mooi verzorgd, inhoud op goed niveau, soms teveel op reken-
machines gericht.

NAAM: COMPU-TECHNIEK
LAND / TAAL: NEDERLAND / NEDERLANDS
PERIODICITEIT: 11 MAAL PER JAAR (niet in augustus)
ADDR. UITGEVER: NANTON PRESS BV., Postbus 93, 3720 AB Bilthoven
Soestdijkseweg 332 N, 3723 HH Bilthoven
ADDR. ABONNEMENTEN: COMPU-TECHNIEK, Postbus 93, 3720 AB Bilthoven
ABONNEMENTEN: Kredietbank Brussel nr. 430-0982931-21 t.n.v. NANTON PRESS
ABONNEMENTSPRIJZEN: 980 Bfr.
PRIJS PER NUMMER: 95 Bfr.
INHOUD: Moeilijk te bepalen na 1 nr.. Klaarblijkelijk meer hobby-gericht
dan DE MICROCOMPUTER. Ook meer Hard.
GEMIDDELDE OMVANG: 50 p. (eerste nr.)
BEOORDELING: Mijns inziens beter dan DE MICROCOMPUTER, maar op basis
van een nr. spreek ik me nog niet te graag uit.

READ

NAAM: COMPUTING TODAY
LAND / TAAL: UK / ENGELS
PERIODICITEIT: MAANDELIJKS
ADDR. UITGEVER: 145 Charina Cross Road, LONDON WC2H 0EE.
ADDR. ABONNEMENTEN: CT Subscription Service, MAP Publications, PO Box 35
Bridge Street, Hemel Hempstead, Herts.
ABONNEMENTSPRIJZEN: 11.00 Pond
PRIJS PER NUMMER: 60 p (0.60 Pond)
INHOUD: Vooral soft nosal PET en TRS-80 gericht
GEMIDDELDE OMVANG: 70-80 p.
BEORDELING: Veelal nuttise, korte programma's of algoritmen

NAAM: CREATIVE COMPUTING
LAND / TAAL: USA / ENGELS
PERIODICITEIT: MAANDELIJKS
ADDR. UITGEVER: Creative Computing, P.O. Box 789-M, Morristown, NJ 07960
ABONNEMENTEN: ** Rechtstreeks uit de states:
1 Jr.: \$15 2 Jr.: \$28 3 Jr.: \$40
foreign surface: + \$9(per Jr.) foreign air mail: + \$30 (per Jr.)
Bovenstaand adres
** dealer Belgie en Nederland
2XF COMPUTERCOLLECTIEF
Attn:F.de Ureeze
Amstel 312A
1017 AP AMSTERDAM, Holland
1 Jr.: f108 2 Jr.: f210 3 Jr.: f300
PRIJS PER NUMMER: \$2.50 (USA)
Ik heb onlangs Creative Computing te koop zien liggen in Belgie: 225 Bf.!!
INHOUD: Soft (spelletjes en educational) + evaluaties .
Zeer weinig hard
GEMIDDELDE OMVANG: 200-250 p.
BEORDELING: Veel zeer goede kant en klare BASIC programma's
deselijke algemene info

NAAM: DATABUS
LAND / TAAL: NEDERLAND / NEDERLANDS
PERIODICITEIT: 10 MAAL PER JAAR
ADDR. UITGEVER: Kluwer Technische Tijdschriften, Van Putlei 33, 2000 Antwerpen
ABONNEMENTEN: Kredietbank nr. 408-0012005-42
ABONNEMENTSPRIJZEN: 1205 Bfr. (incl 6% BTW)
PRIJS PER NUMMER: 135 Bfr.
INHOUD: Hard en soft. De nadruk list echter meer op een oppervlakkige
algemene informatie .
GEMIDDELDE OMVANG: 50-60 p.
BEORDELING: Nosal oppervlakkig, wel mooi verzord.

NAAM: HCCNIEUWSBRIEF
LAND / TAAL: NEDERLAND / NEDERLANDS
PERIODICITEIT: 10 MAAL PER JAAR
ADDR. UITGEVER: Hobby Computer Club, 2250 AC Voorschoten, NL.
Geen vast adres : contactadres bij de bestuursleden
ABONNEMENTEN: Door lid te worden van de HCC
HCC, Leden Administratie, Prinsenhof 11, 2641 RN Pijnacker, NL.
ABONNEMENTSPRIJZEN: 375,-Bfr. PCR.:000-1144548-45
Is tevens lidmaatschap voor 1980 + alle nieuwsbrieven van 1980 .
PRIJS PER NUMMER: ??????
INHOUD: HCCalgemeen en regio nieuws, usersoproepeninfo, hard, soft, enz.
GEMIDDELDE OMVANG: 30-50 p.
BEOORDELING: Zeer deeselijk, nogal veel clubnieuws maar dat is uiteindelijk
het doel van de HCCN. Dikwijls interessante artikels.

NAAM: INTERFACE AGE
LAND / TAAL: USA / ENGELS
PERIODICITEIT: MAANDELIJKS
ADDR. UITGEVER: McPheters, Wolfe & Jones, 16704 Marquardt Ave., Ceritos
Ca 90701, USA
ABONNEMENTEN: ** Rechtstreeks uit de States
1 Jaar: \$28.00
Bovenstaand adres
** INTERFACE AGE EUROPE
Dahlienstr. 4, D-8011 Munchen-Vaterstetten, West Germany
1 Jaar: \$28.00 (56.00 DM) surface mail \$50.00 (92.50 DM) air mail
PRIJS PER NUMMER: \$2.50 (USA)
INHOUD: Algemene artikels + soft + tutorials
GEMIDDELDE OMVANG: 140-150 p.
BEOORDELING: Veel bruikbare programma's en vooral ideeën. De andere artikels
zijn aanspreekbaar om te lezen

NAAM: LIVERPOOL SOFTWARE GAZETTE
LAND / TAAL: UK / ENGELS
PERIODICITEIT: TWEEMAANDELIJKS
ADDR. UITGEVER: Liverpool Software Gazette Limited, 14 Castle Street
Liverpool L2 0TA
ABONNEMENTEN: zelfde adres als boven
ABONNEMENTSPRIJZEN: 9 pond voor 12 nummers binnen de UK.
Geen speciale prijzen gequoteerd voor het buitenland.
PRIJS PER NUMMER: 75 pence
INHOUD: 90% soft
GEMIDDELDE OMVANG: 80-100 p. (elke keer dikker !!)
BEOORDELING: Prachtig! Elders onvindbare info vindt u in LSG.

DE HEER ESVELD HEEFT HET HANDBOEK KRITISCH DOORGENOMEN EN MELDT
ONS VOLGENDE RECTIFICATIES:

PROGRAMMA'S

1) REAL TIME CLOCK op blz 141

```
7 COLORT 8 0 9 0 verbetert op sommige tv's de leesbaarheid
130 DATA .....#C2,#41,#03,.....
210 DATA .....#C3,#57,#03,.....
```

nvdr: in dit nummer vindt U een nieuwe, leesbare versie met toelichtingen.

2) ROTATING PYRAMIDE op blz 143

```
150 ON A-ASC("0") GOTO 500,510,600,610,700,710,5 (5 voor volgende figuur)
2999 DATA 8,12 vervangen door 4005 DATA 8,12
4000 DATA 1,2  vervangen door 4100 DATA 1,2
4200 DATA 6,8  toevoegen
```

nvdr: ASC("0") kan beter vervangen worden door 48.
om beide figuren te maken moet er eerst een CLEAR gebeuren
een uitdaging: roteer de figuur een tijdje in alle richtingen
en probeer dan de originele stand terug te vinden.....


3) DEMOS POKE ADRESSES op blz 159

```
43 IF.....THEN 46
44 .....PROGRAM":GOTO 40
regels 45,64005 en 64006 weglaten
3005 .....:GOTO 3000
64010 .....:P
2041 + vervangen door :
```

nvdr: het gelijknamige programma in de DAI namic-bibliotheek is
een gewijzigde, uitgebreide versie.

4) MUSIC TUTOR op blz 170,171

```
10015 IF.....:GOSUB 6000
```



CATALOG

DE PROGRAMMABIBLIOTHEEK.....

1) INFORMATIE & DEMONSTRATIEPROGRAMMA'S

*

TITEL	GEHEUGEN	MODE	LISTING
- 4 COLOR DEMO (LISSAJOUS)	32K	6	N0
- 16 COLOR/SOUND DEMO	8K	3A	
- OLDTIMER	32K	6	N0
- INFO VIDEORAM MODE0 (DAI)	12K	0	
- THE STING	32K	0	
- MENUET V. BEETHOVEN	32K	0	
- DE VOGELHANDELAAR	32K	0	
- LANDSCHAP	32K	5A	
- PROPELLER (DAI)	32K	6A	N1
- ELECTIONS	32K	5	
- SUM (64 DIGITS-REKENEN) (DAI)	8K	0	
- ROTATING PYRAMIDE (DAI)	32K	6	
- DUTCH DEMO (DAI)	32K	6/5	
- EASY CIRCLES	32K	5A	
- HANEN1	8K	2	
- HANEN2	12K	4	
- HANEN4	32K	6	
- BLOKV1 (COLORG-DEMO)	8K	2	N1
- BLOKV2 (COLORG-DEMO)	12K	4	N1
- BLOKV4 (COLORG-DEMO)	32K	6	N1
- LUNAR LANDING (H. BAKKER)	8K	1A	
- RAKETSPEL (H. BAKKER)	8K	1A	N1
- KANONSPEL (H. BAKKER)	8K	1A	
- SUBMARINE (H. BAKKER)	8K	1A	
- STAR-TREK (H. BAKKER)	8K	1A	
- GETALCONVERSIES (J. VERDONK)	8K	0	N1
- OTHELLO (J. VERDONK)	32K	4A	
- OTHELLO(francais) (JCC)	32K	4A	
- STOCK EXCHANGE (JCC)			
- COLOR-RUIT DEMO (J. VERDONK)	8K	3	N1
- WARI (H. KOOTEN)	8K	0	
- DIG. ANAL. CLOCK (")	8K	0/GRAFISCH	N2
- VIDEOTEXT (H. KOOTEN)	8K	0	
- REACTIESNELHEID (H. KOOTEN)	8K	0	
- YATZEE (ESVELD Jr.)			

NIEUW
NIEUW

NIEUW


NIEUW

NIEUW

NIEUW

NIEUW

NIEUW



CATALOG

PROGRAMMABIBLIOTHEEK Part 2

2) GAMES - EDUCATIEF - UTILITIES

**

TITEL	minimum	GEHEUGEN	MODE	LISTING
- REAL TIME CLOCK		ML	0	N2
- TOWERS OF HANOI		12K	3A	
- ARRANGING NUMBERS + COUNT WITH ME		32K 12K	6 4	
- A MAZE ... AMAZING		48K	5	
- TEKST IN GRAFISCHE MODE		32K	ALLE GRAFISCHE MODES	
- TRAFFICTEST		32K	5	
- MUSIC TUTOR 1 (DA1)		32/48	3/5	
- MUSIC TUTOR + THE STING		48K	6	
ZIE BESCHRIJVING IN DIT NUMMER				
- SIMON		12K	3A	
- VIER OP EEN RIJ		12K	4A	N0
- DISASSEMBLER		32K	0	
- HANNIBAL 2000		32K	6	
- QUICK DESIGN 555		48K	6	
- FAST GRAPHICS		8K	1..6	
MAAK ZONDER PROBLEMEN TEKENINGEN EN ANIMATIES IN MACHINETAAL AAN SUPERSNELHEDEN. KENNIS VAN 8080 ML IS NIET VEREIST. WORDT GELEVERD MET HANDLEIDING EN BASIC DEMO.				

3) PROFESSIONELE PROGRAMMA'S

- ASSEMBLER+EDITOR+LOADER+ DISASSEMBLER	32K	0	2250 FR
BIJ DE ASSEMBLER ONTVANGT U EEN UITVOERIGE HANDLEIDING EN EEN HANDIGE 8080 REFERENCE CARD.			
- GRAFIC ARTIST	48K	5/6	1250 FR
ZIE BESCHRIJVING EN HANDLEIDING IN DIT NUMMER			
- MUSIC ARTIST	48K	6	1250 FR
- WORD PROCESSOR	32K	0	1250 FR
- SORT (DATABANK)	32K	0	1250 FR
- ADRESBESTAND	32K	0	1250 FR
- FAST GRAF TEXT(MACHINETAAL)	8K	1..6	1250 FR
DIT IS DE GRAFTEXT UIT HET HANDBOEK P.154... IN MACHINETAAL. DE SNELHEID BENADERT DE SNELHEID VAN PRINT IN MODE 0 !!! KAN MET BESTAANDE PROGRAMMA'S GEBRUIKT WORDEN ZONDER MERGING. EEN DEMONSTRATIEPROGRAMMA DAT TOONT HOE U FGT KAN GEBRUIKEN IN UW PROGRAMMA'S WORDT BIJGELEVERD. OP AANVRAAG ONTVANGT U TEVENS HET SOURCEPROGRAMMA VAN FGT. beschikbaar vanaf begin december.			

CATALOG

H A N D L E I D I N G DAINAMIC DISASSEMBLER

=====

DOEL DISASSEMBLER

Dit programma laat toe machinetaalprogramma's te bestuderen waarvan we geen dokumentatie bezitten (source list ontbreekt). Dit is van toepassing zowel voor de systeemprogramma's in de ROM als voor machinetaalprogramma's die vooraf in de RAM geladen zijn.

MOGELIJKHEDEN

Na het ingeven van een startadres en drukken van de spacebar wordt 1 lijn gedisassembleerd. Dwz de eerste byte wordt vertaald in een mnemonic en indien een operand voorzien is, dan wordt deze eveneens uitgeprint (single of double byte). De adresteller wordt opgehoogd zodat hij naar de volgende instructiecode wijst. Bij drukken van de spacebar wordt telkens een volgende lijn geprint.

Een machinetaalprogramma kan ook tabellen bevatten die niet in mnemonics vertaalbaar zijn. Hiervoor is een mogelijkheid ingebouwd om byte per byte te printen ; indien mogelijk wordt ook het overeenstemmende ASCII karakter geprint.

Zoals in een assembler kunnen konstanten vervangen worden door symbolen. Komt deze konstante voor als adres dan wordt achter het hexadecimale adres tevens het symbool afgedrukt (=label). Een operand met een waarde die vooraf als symbool gedefinieerd is wordt bij het printen vervangen door dat symbool.

Als we een grote lijst symbolen ingetypt hebben, dan kunnen we deze lijst op cassette plaatsen (voor later gebruik).

KOMMANDO'S

Bij RUN heeft de disassembler enige tijd nodig voor het inlezen van data. Vervolgens wordt naar een startadres gevraagd. Dit startadres kan op 2 manieren ingegeven worden :

- hexadecimaal : vooraf laten gaan door : ,bv :C000
- symbolisch : type gewoon het symbool in (eerst definieren !)

Na car ret wacht de disassembler op een kommando. Het wachten op een kommando wordt altijd aangeduid door de cursor volledig links.

CATALOG

Overzicht kommando's :

- spacebar : disassembleer 1 lijn, dwz
 - print adres hex
 - print bytes hex
 - (print symbool adres)
 - test of eerst byte toegelaten instructiekode is
 - ja : print mnemonic + operand (eventueel symbool)
 - nee: print "invalid opcode"
- T : print 1 byte en ASCII karakter
- S : definieer symbool :
 - type symbool in + car ret
 - de disassembler vraagt nu : EQU ?
 - type nu de hexadecimale waarde inals we proberen een symbool een 2de maal te definieren, dan wordt EQU eerste waarde NEW EQU ? geprint ; er zijn nu 3 mogelijkheden :
 - nieuwe hexwaarde intypen, waardoor 1ste vervalt
 - "-" intypen : het symbool wordt gewist
 - "^" intypen : 1 ste waarde blijft behouden
- L : de lijst van de ingetypte symbolen wordt geprint
- C : ga naar de tape routines
 - kies W/R (schrijf/lees)
 - schrijven : type NAAM in, druk spacebar
 - lezen : altijd zonder naam
- cursor up: verminder het adres met 1 (backspace)
- elke andere toets : een nieuw startadres wordt gevraagd

Telkens we een adres, symbool of hexwaarde moeten intypen kunnen we dit onderbreken door "^", car ret.

OPMERKINGEN

Bepaalde gedeelten van de ROM zijn enkel bereikbaar via bank switching. Deze 4 banks die de adressen :E000 tot :EFFF gebruiken, zijn te disassembleren door resp niets, 1,2 of 3 voor het adres te plaatsen. Voorbeeld : :2E060 adres op 3de bank

Lijn 3 van het BASIC programma bevat DATA 100 . Het getal bepaalt het maximaal aantal symboloen dat gedefinieerd kan worden. Dit kan verlaagd of verhoogd worden om resp minder geheugenruimte te gebruiken of meer symbolen toe te laten. De CLEAR wordt automatisch aangepast. Deze faciliteiten gelden enkel voor de versies vanaf november.

uit PERSONAL COMPUTER WORLD

BM1 300 PRINT "S" 400 FOR K=1 TO 1000 500 NEXT K 700 PRINT "E" 800 END	BM6 300 PRINT "S" 400 K=0 430 DIM M(5) 500 K=K+1 510 A=K/2*3+4-5 520 GOSUB 820 530 FOR L=1 TO 5 540 NEXT L 600 IF K<1000 THEN 500 700 PRINT "E" 800 END 820 RETURN
BM2 300 PRINT "S" 400 K=0 500 K=K+1 600 IF K<1000 THEN 500 700 PRINT "E" 800 END	BM7 300 PRINT "S" 400 K=0 430 DIM M(5) 500 K=K+1 510 A=K/2*3+4-5 520 GOSUB 820 530 FOR L=1 TO 5 535 M(L)=A 540 NEXT L 600 IF K<1000 THEN 500 700 PRINT "E" 800 END 820 RETURN
BM3 300 PRINT "S" 400 K=0 500 K=K+1 510 A=K/K*K+K-K 600 IF K<1000 THEN 500 700 PRINT "E" 800 END	BM8 300 PRINT "S" 400 K=0 500 K=K+1 530 A=K↑2 540 B=LOG(K) 550 C=SIN(K) 600 IF K<100 THEN 500 700 PRINT "E" 800 END
BM4 300 PRINT "S" 400 K=0 500 K=K+1 510 A=K/2*3+4-5 600 IF K<1000 THEN 500 700 PRINT "E" 800 END	
BM5 300 PRINT "S" 400 K=0 500 K=K+1 510 A=K/2*3+4-5 520 GOSUB 820 600 IF K<1000 THEN 500 700 PRINT "E" 800 END 820 RETURN	

	BM1	BM2	BM3	BM4	BM5	BM6	BM7	BM8
Altos ACS 800-2	1.4	4.3	11.3	11.3	12.0	21.2	34.9	2.7
TRS-80 Model II (Int)	1.0	4.0	13.0	13.0	14.0	20.0	30.0	6.0
(S/prec)	1.0	5.0	13.0	13.0	14.0	23.0	35.0	6.0
(D/prec)	—	6.0	41.0	43.0	44.0	52.0	65.0	7.0
Periflex 630/48	4.5	10.5	27.5	28.5	31.5	59.0	79.5	60.0
Acorn Atom (Int)	0.8	5.5	10.0	11.5	14.5	20.0	31.5	—
(F/P)	—	—	30.5	27.0	30.0	—	—	26.0
DDE SPC/1	4.8	6.2	14.7	13.9	14.7	41.1	58.1	2.6
SuperBrain	1.6	5.2	14.0	13.9	14.8	26.3	43.2	5.6
BASF 7120	2.4	7.0	35.0	36.5	39.0	50.0	63.0	114.0
Atari 400/800	2.3	7.4	19.9	23.2	26.8	40.7	61.5	43.1
DAI	0.9	4.8	10.1	9.8	11.2	18.1	30.1	2.1
CBM 8032	1.7	10.0	18.4	20.3	21.9	32.4	51.0	11.9

Int = Integer Basic
 F/P = Floating point Basic

S/prec = Single precision
 D/prec = Double precision

All timings in seconds.

LIST

LIST-PRINT

Doel :

Het maken van een listing van een BASIC-programma op de matrix printer, met op het einde van elke pagina de naam van het programma en een paginanummer.

Restricties :

- 1) Het te listen programma mag de regelnummers 65400 en hoger niet gebruiken.
- 2) Het aantal characters per regel (exklusief het regelnummer) mag maximaal 74 zijn.
- 3) Bedoeld voor papier waar 66 regels per pagina op kunnen. Aan te passen door de waarden van TT en Y te veranderen in de regels 65460, 65490, 65500 en 65505.

Werkwijze :

Merse het te listen programma en LIST-PRINT (L-P) volgens de methode zoals beschreven in het DAI-handbook, pagina 59, par. 6.2.4.4.

RUN 65400

L-P vraagt naar de naam van het te listen programma en naar het nummer van de eerste pagina van de te maken listing (hetseer de mogelijkheid geeft door te nummeren). Hierna geeft L-P enige instructies betreffende het instellen van de printer.

Het duurt enige tyd voordat het printen start, tevens treedt er telkens een korte pauze op na het printen van het paginanummer.

Als laatste wordt geprint :

```
65400 REM ***** E I N D E L I S T I N G *****
```

```
          'NAAM PROGRAMMA'      Pagina 'XXX'
```

Hierna gaat L-P in de loop '65220 GOTO 65220', waardoor nu de printer afgeschakeld kan worden zonder dat er op de volgende pagina ekstra printout komt.

LIST-PRINT

```

65400 REM ***** E I N D E L I S T I N G *****
65405 PRINT CHR$(12):PRINT :PRINT :PRINT :PRINT
65410 PRINT "LIST-PRINT (max. reegelnummer is 65399).":PRINT :PRINT :PRINT
65415 PRINT "Naam van het programma ?"
65420 INPUT N$:PRINT :PRINT :PRINT "Nummer eerste pagina ?":INPUT P$:PRINT
65425 PRINT :PRINT :PRINT "Zet de vouw van het papier selyk"
65430 PRINT "met de bovenzijde van het PRINT-HEAD."
65435 PRINT :PRINT "Schakel de Printer aan en druk "
65440 INPUT "op de RETURN-toets.":NN$:PRINT
65442 PRINT :POKE #FF06,#E
65443 PRINT N$:PRINT :S=0.0
65445 A=PEEK(#2A0):AA=PEEK(#29F):P=A*256.0+AA:Q=P
65450 L=PEEK(Q):A=PEEK(Q+1):AA=PEEK(Q+2):R=A*256.0+AA
65455 IF R=65500.0 THEN 65460:Q=Q+L+1.0:GOTO 65450
65460 TT=54.0:ADRA=Q+4.0:ADRB=Q+6.0:Q=P
65465 L=PEEK(Q):POKE ADRA,PEEK(Q+1):POKE ADRA+1,PEEK(Q+2)
65470 T=T+1.0:Q=Q+L+1.0:L=PEEK(Q)
65475 IF PEEK(Q+1)=#FF AND PEEK(Q+2)=#78 THEN GOTO 65495
65480 IF T=TT THEN 65485:GOTO 65470
65485 POKE ADRB,PEEK(Q+1):POKE ADRB+1,PEEK(Q+2):GOSUB 65500
65490 T=0.0:TT=59.0:Q=Q+L+1.0:GOTO 65465
65495 S=1.0:PRINT :GOTO 65485
65500 LIST 1-65400:IF TT<>54.0 THEN GOTO 65505:PRINT :GOTO 65510
65505 V=59.0-T:FOR X=1.0 TO V:PRINT :NEXT X
65510 PRINT TAB(50):N$:"          Pagina":P$:P%=P%+1
65515 PRINT :PRINT :PRINT :PRINT :IF S=1.0 THEN 65520:RETURN
65520 GOTO 65520
65525 END

```

ZAKDOEK LEGGEN

```

5   COLORG 12 3 15 3
7   MODE 6
9   FOR A=YMAX/8.0 TO YMAX/140.0 STEP -2.0
10  FOR I=0.0 TO 2.0*PI STEP PI/150.0
20  R=A*(3.0-COS(6.0*I))
30  X=R*COS(I)
40  Y=R*SIN(I)
50  DOT X+XMAX/2,Y+YMAX/2 3
60  NEXT I
65  NEXT A
67  FOR S=1.0 TO 3.0
70  FOR A=1.0 TO 15.0
80  COLORG 0 A 12 3
95  WAIT TIME 100
96  NEXT A
97  NEXT S
100 GOTO 100

```